Approved:

D. H. Adams

CSC INTEROPERABILITY SYSTEMS INTERNATIONAL S.A.

WORKBOOK ON ADA PROGRAMMING

CSC INTEROPERABILITY SYSTEMS INTERNATIONAL S.A.
104 IMITTOU, 16561 GLYFADA, GREECE.

The purpose of this workbook is to teach you about Ada in an overview fashion and is by no means a complete tutorial. This workbook is provided as an aid to understanding the Ada language through numerous examples and is not intended as a replacement for a good textbook on Ada.

This workbook is based on:

Workbook
on
Ada Programming
Version 1.0

Author:
Richard Conn
University of Cincinnati
Department of Electrical and Computer Engineering
March, 1992

Modifications by H.Winter, ARC ISI, October 1993:

Some minor modifications have been done to convert the document from "words for windows" format. At the same time all examples have been reworked to meet the company coding standard.

Appendix A (listing of Text_io package), Appendix B (Suggested Reading) have been removed. The solutions in Appendix C have been moved into a separate file.

Table of Contents

1 OVERVIEW AND BASICS

1.1 A SAMPLE ADA PROGRAM

Figure 1. A Sample Ada Program

```
with
  Text_IO;  -- context specification

procedure Count_Down is  -- an Ada mainline is always a procedure

  Number_of_Iterations : Integer := 1;
    -- local variable definition

  package Int_IO is new Text_IO.Integer_IO(Integer);
    -- local package for Integer I/O

begin -- Count_Down
  Text_IO.Put("Enter number of iterations: ");
  Int_IO.Get(Number_of_Iterations);
  Text_IO.New_Line;
  Text_IO.Put("The loop will run for ");
  Int_IO.Put(Number_of_Iterations);
  Text_IO.Put_Line(" iterations");

  for Index in 1 .. Number_of_Iterations loop
    Text_IO.Put("Iteration: ");
    Int_IO.Put(Index, 4);  -- 4 column field
    Text_IO.New_Line;
  end loop;

end Count_Down;
```

A few notes about this program:

- Comments are denoted by a double-dash (--) and extend to the end of the line.

- The mainline is a procedure subprogram with no arguments. A subprogram is one of the four program units in Ada.

- The context specification indicates other Ada program units which are required by the current program unit. This procedure uses the package Text_IO for input and output support. Text_IO is an Ada package (a package is another kind of program unit) which contains many routines for inputting and outputting characters, strings, integers, etc.

- Number_of_Iterations is a variable of type Integer. The predefined type Integer is defined in the package called Standard, and Standard is automatically included in the context of all Ada program units so it need not be specified in a with statement.

- Int_IO is a package within this procedure that is created from a generic package called Integer_IO which is contained within the predefined package Text_IO. A generic unit (generic units may be subprograms or packages) is a third kind of Ada program unit.

- Most of the code in this program revolves around displaying and inputting text and integers. Note that the output is viewed as a stream, and the Put function places something into that stream. A new line is never assumed and must be explicitly placed into the stream via a Text_IO.New_Line procedure call or a Text_IO.Put_Line procedure call.

- Variables must be declared in Ada before they are used, but variables in a for loop are automatically created. The type of these variables is determined from the types of the parameters used. In this case, Index is of type Integer because Number_of_Iterations is of type Integer.

Problem 1.1: Type in the above program, compile it, and run it. Try giving various integer values, and be sure to include negative values. Try giving it erroneous inputs also, such as a character other than an integer.

1.2 LEXICAL ELEMENTS

There are 4 kinds of lexical elements, the basic language components of Ada:

- Identifiers -- names that identify various Ada objects, such as variables, constants, procedures, functions, and so forth. Identifiers begin with a letter and contain letters, digits, and underscore characters. Examples:

```
A
Number_of_Iterations
AGE
P23
This_is_a_very_long_identifier
```

There are seven rules to follow when constructing Ada identifiers:

1. Identifiers can only contain letters, digits, and underscores.
2. Identifiers can be of any length as long as the entire identifier appears within a single line.
3. Ada does not distinguish between upper- and lower-case letters.
4. Identifiers cannot have the same name as a reserved word.
5. Identifiers must begin with a letter.
6. Identifiers cannot have underscores at the beginning, at the end, or adjacent to each other.
7. Underscores in identifiers are significant.

- Delimiters -- symbols that have a special meaning within Ada. They are often used as operators and statement terminators, but there are other purposes as well.

  The single-character delimiters are:

```
+  -  *  /  =  <  >  &  |  ;  #  (  )  .  :  "  '
```

  The double-character delimiters are:

```
**  /=  <=  >=  :=  <>  =>  <<  >>  ..
```

- Literal -- a particular value of a type that is explicitly written and not represented by an identifier. Kinds of literals are:

  Integer literals:
  ```
  0   187   2_546   2e3   12E+15   2#0010#    16#fc#
  ```

  Real literals:
  ```
  0.0   127.021   4_728.001_002   16#f.2c#    1.2e35
  ```

  Character literals:
  ```
  'a'   'A'      '''   '8'   '*'
  ```

  String literals:
  ```
  "This is a string"  """"  """Hello, Jack"", he said"
  ```

- Comment -- a string of characters beginning with a double dash and extending to the end of the line. Comments are ignored during compilation.

Problem 1.2: Write an Ada program that displays a single quote and a double quote.

1.3 ADA PROGRAM UNITS

An Ada source file contains one or more Ada program units. An Ada Program or System is composed of one or more program units, where a program unit is a subprogram, a package, a task, or a generic unit. Each program unit is divided into two parts: (1) a specification, which defines its interface to the outside world, and (2) a body, which contains the code of the program unit.

When an Ada compiler runs, it compiles an Ada source file. The compiler places the program units from the source file into an Ada Program Unit Library if they compile successfully. Later, when an Ada linker (also commonly called a binder) is run, and executable is produced from program units within an Ada Program Unit Library. The mainline is specified to the linker, and the linker sets this program unit up to begin execution when the program is run. An Ada mainline is always an Ada procedure (a subprogram unit). Program units may be nested -- they can contain zero or more other program units.

1.3.1 Subprogram Program Units

A subprogram is an expression of sequential action. Two kinds of subprograms exist: a procedure and a function.

Figure 2. Sample Procedure Subprogram

```
type REAL_ARRAY_t is array (1 .. 20) of FLOAT;
procedure SORT(Items : in out REAL_ARRAY_t);    -- specification
procedure SORT(Items : in out REAL_ARRAY_t) is -- body
  -- definitions of types, objects, exceptions, and other program units
  -- (subprograms, packages, tasks, and generic units) local to SORT
begin -- SORT
  -- body of code which implements SORT
  null;  -- no code for now
end SORT;
```

Figure 3. Sample Function Subprogram

```
function Is_Zero(Item : in FLOAT) return BOOLEAN;    -- specification
function Is_Zero(Item : in FLOAT) return BOOLEAN is -- body
  -- definitions of types, objects, exceptions, and other program units
  -- local to Is_Zero
begin -- Is_Zero
  null;  -- no code for now
  return TRUE;
end Is_Zero;
```

1.3.2 Package Program Units

A package is a collection of computational resources, including data types, data objects, exception declarations, and other program units (subprograms, tasks, packages, and generic units). A package is a group of related items.

In object-oriented programming, a package contains the definition of a particular object or a class of objects. This includes the member data and all methods associated with the object or class.

Packages are fundamental to Ada. For instance, Ada by itself has no Input or Output capabilities, so the package Text_IO is provided with all Ada compilers to provide input and output capabilities for characters, strings, floating point numbers, fixed point numbers, integers, and user-defined types.

An example of an Ada package specification is:

Figure 4. Sample Package Specification

```
package Console_Terminal_Screen is
  -- definitions of types, objects, exceptions, and other program units
  -- provided to external program units by this package
  subtype ROW_t is INTEGER range 1 .. 24;
  subtype COLUMN_t is INTEGER range 1 .. 80;
  procedure Clear_Screen;
  procedure Position_Cursor(At_Row : in ROW_t; At_Column : in COLUMN_t);
  procedure Write(Item : in STRING);
end Console_Terminal_Screen;
```

An example of an Ada package body which goes with this specification is:

Figure 5. Sample Package Body

```
package body Console_Terminal_Screen is
  -- definitions of types, objects, exceptions, and other program units used
locally
  -- by this package
  procedure Clear_Screen is
  begin -- Clear_Screen
    -- code to implement Clear_Screen
  end Clear_Screen;

  procedure Position_Cursor(
    At_Row    : in ROW_t;
    At_Column : in COLUMN_t)
  is
  begin -- Position_Cursor
    -- code to implement Position_Cursor
  end Position_Cursor;

  procedure Write(Item : in STRING) is
  begin -- Write
    -- code to implement Write
  end Write;

end Console_Terminal_Screen;
```

1.3.3 Task Program Units

A task is an expression of action implemented in parallel with other tasks. A task is a body of code which may be implemented on one processor, a multiprocessor (more than one CPU), or a network of processors. Like other program units, a task has a specification and a body.

An example of a task specification is:

Figure 6. Sample Task Specification

```
task Terminal_Driver is
  -- entry points to the task specify how other tasks communicate
  -- with this task
  entry Get(Char : out CHARACTER);
  entry Put(Char : in CHARACTER);
end Terminal_Driver;
```

An example of a task body with matches the above specification is:

Figure 7. Sample Task Body

```
task body Terminal_Driver is
  -- local data, etc.
begin
  accept Get(Char : out CHARACTER) do
    -- code which implements the Get entry
  end Get;
  accept Put(Char : in CHARACTER) do
    -- code which implements the Put entry
  end Put;
end Terminal_Driver;
```

1.3.4 Generic Program Units

A generic unit is a special implementation of a subprogram or package which defines a commonly-used algorithm in data-independent terms. Generic units are reusable software components which contain algorithm definitions.

An example of a generic subprogram specification and body is:

Figure 8. Sample Generic Specification

```
generic -- specification
  type ELEMENT_t is private;  -- thing manipulated
procedure Exchange(Item1 : in out ELEMENT_t; Item2 : in out ELEMENT_t);

procedure Exchange(Item1 : in out ELEMENT_t; Item2 : in out ELEMENT_t) is
  -- body
  Temp : ELEMENT_t;
begin -- Exchange
  Temp := Item1;
  Item1 := Item2;
  Item2 := Temp;
end Exchange;
```

1.3.5 Procedures as Main Programs

Ada does not have a separate construct for a main program. Instead, Ada program units (subprograms, packages, tasks, and generic units) are compiled into an Ada Program Unit Library and then, at some later time, one of the procedures is selected to be the mainline procedure at which execution of the program is to begin. A mainline procedure has no parameters.

1.3.6 Ada Program Unit Library

The Ada compiler outputs into the current Ada Program Unit Library as its only target. The compiler does not necessarily create .o files like a C + + compiler may. Reuse is accomplished by accessing existing Ada program unit libraries, thereby gaining use of the program units contained in them.

All Ada compilers provide a common program unit library that contains the following packages:

```
package STANDARD -- contains integers, floats, and operations on them
package TEXT_IO -- support for I/O
package SYSTEM -- ability to address memory
package SEQUENTIAL_IO -- support for sequential I/O only
package DIRECT_IO -- support for random I/O only
package IO_EXCEPTIONS -- errors which may come up with I/O
package LOW_LEVEL_IO -- special platform-specific I/O
```

Creating an Executable. The Ada Binder builds an executable from a procedure that is located any where in the Ada Program Unit Libraries. A chain of program units is assembled to create this executable:

- The mainline procedure comes first, and this procedure requires certain program units for support (it depends upon these program units and they are named in its with statements).

- The program units withed by the mainline procedure are incorporated into the executable, and these program units may with other program units.

- The next layer of program units is loaded, and so on as they with yet other program units.

- The Ada runtime system, which supports initialization, exception handling, tasking, and other features of the language, may either be included in the executable or tied into by the executable. The with statement in Ada causes one program unit to gain access to another.

1.3.7 Problem

> Problem 1.3: Write an Ada program that demonstrates a package, a procedure, and a function.

1.4 BASIC AND EXTENDED CHARACTER SETS

The Basic Character Set (BCS) is one of two character sets used by Ada programs. The BCS was designed to facilitate transportability between computer systems. The BCS consists of:

uppercase letters only: A - Z
digits: 0 - 9
special characters: " # ' ( ) * + - / , . : ; < = > _ | &
the space character

The Extended Character Set (ECS) maps to the 95-character ASCII (American Standard Code for Information Interchange) set:

all characters in the BCS
more special characters: ! ~ $? @ [ ] { } \ ' ^ %
lowercase letters: a - z

Package ASCII. Within the supplied package STANDARD is another package, package ASCII. Package ASCII provides two things: (1) names for the non-printing ASCII characters and (2) names for the characters in the ECS which are not a part of the BCS. Examples:

Figure 9. Sample definitions with package ASCII

```
C1  : Character := ASCII.NUL;
C2a : Character := '#';
C2b : Character := ASCII.SHARP;  -- same as C2a
C3a : Character := 'a';
C3b : Character := ASCII.LC_A;   -- same as C3a
```

The predefined package STANDARD is the only Ada package which is automatically withed by every Ada program unit without the programmer having to explicitly do so. Consequently, package ASCII is always available. A sample program:

Figure 10. Sample ASCII Demo

```
with
  Text_IO;  -- for output
procedure ASCII_Demo is
begin -- ASCII_Demo
  Text_IO.Put("Ring the Bell: ");  -- Put for a string
  for I in 1 .. 20 loop
    Text_IO.Put(ASCII.BEL);  -- Put for a character
  end loop;
  Text_IO.New_Line;
end ASCII_Demo;
```

Problem 1.4: Write an Ada program that displays the numeric values of the ASCII characters from ASCII.NUL to ASCII.SUB.

1.5 RESERVED WORDS

Reserved words are identifiers which may be used in only certain contexts. They may not be used as variables, enumeration literals, procedure names, etc. They may be a part of strings (e.g., "my package is in"). They may be a part of other lexical units (e.g., PACKAGE_52 is OK).

Figure 11. Complete List of Ada Reserved Words

abort	**declare**	**generic**	**of**	**select**
abs	**delay**	**goto**	**or**	**separate**
accept	**delta**		**others**	**subtype**
access	**digits**	**if**	**out**	
all	**do**	**in**		**task**
and		**is**	**package**	**terminate**
array	**else**		**pragma**	**then**
at	**elsif**	**limited**	**private**	**type**
	end	**loop**	**procedure**	
begin	**entry**			**use**
body	**exception**	**mod**	**raise**	
	exit		**range**	**when**
case		**new**	**record**	**while**
constant	**for**	**not**	**rem**	**with**
	function	**null**	**renames**	
			return	**xor**
			reverse	

Problem 1.5: Write an Ada program which uses the reserved words elsif, for, and in.

1.6 PACKAGE STANDARD

As mentioned above, package STANDARD is automatically withed and used by all Ada program units. Package STANDARD contains:

- type **BOOLEAN** and the associated operations:
  = /= < <= > >= **and** **or** **xor** **not**

- type **INTEGER** and the associated operations:
  = /= < <= > >=
  + - **abs** (unary operations)
  + - * / **rem** **mod** (binary operations)
  **

- type **FLOAT** and the associated operations:
  = /= < <= > >=
  + - **abs** (unary operations)
  + - * / (binary operations)
  ** (INTEGER exponent)

- the types universal real, universal integer, and universal fixed along with their operations:

  universal real := universal integer * universal real
  universal real := universal real * universal integer
  universal real := universal real / universal integer
  universal fixed := user-defined fixed * user-defined fixed
  universal fixed := user-defined fixed / user-defined fixed

- types **CHARACTER** and **STRING** and their associated operations:
  = /= < <= > >=
  & (binary operation for both CHARACTER and STRING)

- subtype **NATURAL** is **INTEGER** range 0 .. INTEGER'LAST;
- subtype **POSITIVE** is **INTEGER** range 1 .. INTEGER'LAST;

- predefined exceptions:

```
CONSTRAINT_ERROR
PROGRAM_ERROR
TASKING_ERROR
NUMERIC_ERROR
STORAGE_ERROR
```

Problem 1.6: Write an Ada program which builds and displays a truth table showing the results of the logical and, or, and xor of two Boolean variables.

2 TYPES

In Ada, every object belongs to a class of objects. Classes of objects are denoted by a type. A type characterizes:

- a set of values which objects of that type may take on (e.g., integers range in value from INTEGER'FIRST to INTEGER'LAST)

- a set of attributes (e.g., INTEGER'LAST is the last integer)

- a set of operations which may be performed on objects of that type (e.g., the package STANDARD defines the type INTEGER and the operations which may be performed on objects of type INTEGER)

Several classes of types are available in the Ada language:

- scalar data types, which assume a single value at any one time

  integer
  real (both floating point and fixed point)
  enumeration

- composite data types, which assume one or more values in various parts of them at any one time

  array
  record

- access data types, which are pointers to other types of objects (these are handled in a distinctly different fashion from scalar data types, although they do assume a single value at any one time)

- private data types, whose contents are not of interest to the user (employed in conjunction with Ada packages)

- subtypes, which are created from parent types and are compatible with the parent types

- derived types, which are created from parent types and are not compatible with the parent types

This section of the workbook discusses all of these types except for private data types, which are covered in the discussion on Ada packages.

2.1 SCALAR DATA TYPES

Scalar types are types that do not have any lower-level components; scalar types assume a single value at any one time, like an integer.

There are two kinds of scalar types: discrete and real. A discrete type is one whose values have immediate successors and predecessors, such as Boolean, Integers, and user-defined enumeration types. A real type is one whose values are numbers that form a continuum, such as floating point numbers and fixed point numbers.

Scalar types are either predefined or user-defined. The following table presents the predefined scalar data types and the syntax for defining user-defined scalar data types:

Figure 12. Predefined Scalar Data Types

```
Kind of Scalar Data Type                    Explanation
========================                    ========================
Integer:
        INTEGER                             a predefined type
        NATURAL                             a predefined type, >= 0
        POSITIVE                            a predefined type, >= 1

        type INDEX_t is range 1 .. 50;      syntax for creating
                                            user-defined integer-like types

Real (floating point and fixed point):

        FLOAT                               a predefined type

        type MASS_t is digits 10;           a user-defined floating point
                                            type with 10 significant digits

        type VOLTAGE_t is delta 0.01        a user-defined fixed point type
          range 0.0 .. 50.0;

Enumeration:

        BOOLEAN                             a predefined type, values are
                                            FALSE and TRUE

        CHARACTER                           a predefined type, values are
                                            from the Extended Character Set

        type COLOR_t is (RED,               a user-defined enumeration type
          GREEN, BLUE);
```

It is important to be able to distinguish between numeric and discrete scalar data types since only discrete types may be used for loop variables, such as in a for loop.

Figure 13. Numeric and Discrete Types

	Numeric	Discrete
Integer	X	X
Real	X	
Enumeration		X

Universal types, which are of a literal form and may match to any of a number of similar types, exist in Ada. The following classes of universal types exist:

- Universal Integer
  Integer Literals, e.g.: `12`
  Integer Named Numbers, e.g.: `DOZEN_c : constant := 12;`

- Universal Real
  Real Literals, e.g.: `3.14159`
  Real Named Numbers, e.g.: `PI_c : constant := 3.14159_26535;`

In clarification, do not confuse these universal values with typed values, such as:

```
DOZEN_c : constant INTEGER := 12;    -- this is of type INTEGER
DOZEN_c : constant := 12;            -- this is a universal integer
```

There are two main advantages to universal types:

1. Universal types do not have any practical size constraints:
   `SPEED_OF_LIGHT_c : constant := 186_282;  -- valid on even 16-bit machines`

2. Code may execute faster: when named numbers are combined with other named numbers or numeric literals, the resulting expression may be evaluated at compilation time rather than run time.

Expressions consisting of only named numbers or numeric literals are called literal expressions. Named numbers may be initialized to literal expressions but not to non-literal expressions:

```
DOZEN_c : constant := 12;
BAKERS_DOZEN_c : constant := DOZEN_c + 1; -- OK because 'DOZEN+1' is
                                          -- a literal expression
DOZEN_c : constant INTEGER := 12;
BAKERS_DOZEN_c : constant := DOZEN_c + 1; -- not OK because DOZEN is a
                                          -- variable
```

Problem 2.1: Write an Ada program which consists of a procedure inside a package that employs an enumeration type containing the values RED, GREEN, BLUE, YELLOW, ORANGE, TURQUOISE, MAGENTA, and VIOLET in that order. The procedure should contain a for loop that displays the values from RED on up to some limit which is specified by a constant. Use typename'IMAGE to convert from an enumeration type value to a representative string.

2.2 SUBTYPES AND DERIVED TYPES

Subtypes are types created from an existing "parent" type which are distinct but compatible with the parent. Objects of a subtype may be mixed with objects of the parent type in an expression. For example,

```
subtype SINT_t is INTEGER range 1 .. 10;
I  : INTEGER;
SI : SINT_t;
SI := 5;
I  := 2 + SI;  -- OK because SI is a subtype
```

Derived types are types created from an existing "parent" type which are distinct and separate (incompatible) from the parent. They have the same physical structure but the compiler views them as being distinct and enforces this view by not allowing operations dealing with the parent and the derived types to exist unless they are defined explicitly by the programmer. For example:

```
type SINT_t is new INTEGER range 1 .. 10;  -- this is derived
I  : INTEGER;
SI : SINT_t;
SI := 5;
I  := 10 + SI;  -- will raise an error message at compile time
```

Derived types are different from subtypes:

- A derived type introduces a new type, distinct from its parent.

- A subtype places a restriction on an existing type, but it is compatible with its parent.

Derived types make a lot of sense, providing a check when mapping to the real world. For instance, in the real world, you would never try to add something of type SPEED (say, in miles per hour) to something of type TIME (say, in hours). It simply does not make sense. Derived types in Ada prevent this kind of thing from happening:

```
type SPEED_t is new Integer range 0 .. 2_000;  -- MPH
type TIME_t is new Integer range 0 .. 24;  -- hours
S  : SPEED_t := 25;
T  : TIME_t := 12;
S  := S + 5;     -- legal, no problem
T  := T + 1;     -- legal, we are still within the range constraint
S  := S + T;     -- illegal, flagged at compile time
```

Problem 2.2: Write an Ada program which uses the definitions for SPEED and TIME from above. Add a new derived type for DISTANCE. Create an overloaded "*" operator which multiplies SPEED by TIME to produce DISTANCE. Use this operator to create a table of speeds, times, and distances (2..24 hours, in 2 h increments and 200..2000 MPH in 200 MPH increments)

2.3 ARRAYS

An array is an object that consists of multiple homogenous components (i.e., each component is of the same type). An entire array is reference by a single identifier:

```
type FLOAT_ARRAY_t is array (1 .. 10) of FLOAT;  -- array type declaration
My_Float_Array : FLOAT_ARRAY_t;  -- array object definition
```

Each component of an array is referenced by the identifier which references the array being followed by an index in parentheses. Notice that two types are generally involved: the type of the array elements and the type of the index. Continuing the above example:

```
My_Float_Array(5) := 12.2;  -- assign one element

My_Float_Array(5 .. 7) := (1.0, 2.0, 3.0);
    -- assign three element using a slice

for I in My_Float_Array'FIRST .. My_Float_Array'LAST loop
   My_Float_Array(I) := 0.0; -- init all elements
end loop;
```

The general syntax of an array type declaration is:

`type` ARRAY_TYPE_NAME `is array` (INDEX_SPECIFICATION) `of` ELEMENT_TYPE;

Some notes:

- ARRAY_TYPE_NAME is the name given to this type, not the name of a specific array; specific arrays are declared later as array objects
- INDEX_SPECIFICATION is the type and value range limits, if any, of the index
- ELEMENT_TYPE is the type of the array elements

Some examples of array type declarations and array definitions:

```
type COLOR_t is (RED, GREEN, BLUE); -- an enumeration type (used later)

type VALUES_t is array (1 .. 8) of FLOAT;  -- a vector of 8 real numbers

My_Floats  : VALUES_t;  -- object definition
His_Floats : VALUES_t := (1.1, 2.2, 3.3, 4.4, 5.5, 6.6, 7.7, 8.8);
   -- object definition with initialization

Zero_Floats : VALUES_t := (others => 0.0);  -- initialized to 0.0


type CVAL_t is array (COLOR) of FLOAT;
   -- a vector of 3 real numbers indexed by RED, GREEN, and BLUE

My_Color_Values : CVAL_t;  -- object definition

Reference_Color_Values_c : constant CVAL_t := (1.1, 2.2, 3.3);
   -- constant object definition

Our_Colors  : Cval_t := (RED   => 1.0,
                         GREEN => 2.0,
                         BLUE  => 3.0);
  -- index associations spelled out for readability


type SCREEN_DOTS is array (1 .. 1024, 1 .. 1024) of COLOR;
My_CRT_Screen : SCREEN_DOTS := (others => (others => RED));
    -- all RED
```

An entire array may be initialized by assigning it to an array aggregate. Some examples:

```
type MENU_SELECTION_t is (SPAM, MEAT_LOAF, HOT_DOG, BURGER);
  -- an enumeration naming the different items on the menu

type DAY_t is (MON, TUE, WED, THU, FRI);
  -- an enumeration naming the days of the week

type SPECIAL_LIST_t is array (DAY_t) of MENU_SELECTION_t;
  -- an array type, indexed by the days of the week, of menu items

Specials : SPECIAL_LIST_t;  -- an array

Specials := SPECIAL_LIST_t'(SPAM, HOT_DOG, BURGER, MEAT_LOAF, SPAM);
  -- an array aggregate assigning a value to the array Specials

Specials := (SPAM, HOT_DOG, BURGER, MEAT_LOAF, SPAM);
  -- the qualifier is not needed *
  -- it simply improves readability
                                        * (in this case)
```

```
Specials := (MON => SPAM,
             TUE => HOT_DOG,
             WED => BURGER,
             THU => MEAT_LOAF,
             FRI => SPAM);  -- another improvement on readability

Specials := (MON | FRI => SPAM,
             TUE | WED | THU => BURGER);

Specials := (MON .. WED => BURGER, others => MEAT_LOAF);
```

Some additional notes on arrays:

- Arrays may have as many dimensions as desired.

- So far, array types have been constrained (i.e., the number of elements in the arrays have been determined in advance during the declaration of the array type). In Ada, array types may also be unconstrained, where the objects derived from these types are not constrained until the definitions of these objects. For example:

```
type FLOAT_ARRAY_t is array (NATURAL range <>) of FLOAT;
 -- index is of type NATURAL and is constrained at object definition

My_Array     : FLOAT_ARRAY_t (1 .. 10);  -- 10 elements
His_Array    : FLOAT_ARRAY_t (5 .. 12);  -- 8 elements
Zero_Array_c : constant FLOAT_ARRAY_t := (0.0, 0.0, 0.0);
  -- 3 elements indexed from NATURAL'FIRST (0, 1, 2)
```

- A STRING is an unconstrained array indexed by POSITIVE of CHARACTER objects. The type STRING is predefined in the package STANDARD:

```
type STRING is array (POSITIVE range <>) of CHARACTER;
```

- Once a STRING object has been defined, it may be assigned a value by using array aggregate notation or by using quotes:

```
My_Name : STRING(1 .. 4) := "John";
My_Name := ('J', 'i', 'm', ' ');
```

Problem 2.3: Write an Ada program which uses the following types to create several tic-tac-toe board positions and display them:

```
type SQUARE_t is (EMPTY, CROSS, ZERO);
type ROW_t is (ROW_1, ROW_2, ROW_3);
type COL_t is (COL_1, COL_2, COL_3);
type TIC_TAC_TOE_BOARD_t is array
  (ROW_t, COL_t) of SQUARE_t;
```

2.4 STRINGS

A string is a kind of vector used so often that it is useful to make a special mention of it. Some notes about strings follow.

- A double quote may be placed into a string by using two double quotes in a row:

```
Message_c : constant STRING := """Hi""";
```

- Strings are fixed in their length when they are defined. Ada does not support any predefined variable-length strings, but such strings can be implemented.

```
Pet_Name : STRING(1 .. 5);
Pet_Name := "Pete";     -- illegal because Pet_Name has 5 characters
                        -- and "Pete" has 4

Pet_Name := "Repeat";   -- illegal for similar reasons
Pet_Name := "Pete ";    -- OK (note trailing space)
Pet_Name(1 .. 4) := "Jake";     -- OK due to slice size matching string
                                -- length
```

- Slices can be used to assign parts of a string:

```
Message(1 .. 2) := "Hi";
```

- Although the size of a STRING variable is set by its constraint, the size of a STRING constant may be inferred from the size of the aggregate assigned to it:

```
My_Pet_c : constant STRING := "Jake the Snake";
```

- Package Text_IO supports a STRING input procedure called Get_Line which may be used to input values into STRINGs. Get_Line requires the name of the STRING object and a variable of type NATURAL as parameters:

```
Input_Line : STRING(1 .. 80);
Input_Last : NATURAL;

Text_IO.Get_Line (Input_Line, Input_Last);

-- Assume that the user types "HELLO<CR>", where <CR> is
-- the RETURN key, Input_Line(1..5) = "HELLO" and
-- Input_Last = 5 are the results
```

Problem 2.4: Write an Ada program which inputs five lines from the console using package Text_IO. After the five lines are input, the program displays them back to the user.

2.5 BOOLEAN VECTORS

A boolean vector is a user-defined type which is a vector of BOOLEAN objects:

```
type BOOLEAN_VECTOR is array (POSITIVE range <>) of BOOLEAN;
```

A boolean vector is the only type of array that can be operated on by the logical operators and, or, xor, and not as a whole (i.e., the operators operate on the entire array at one time rather than forcing the programmer to reference each element individually). For example:

```
declare
  type BOOLEAN_VECTOR_t is array (POSITIVE range <>) of BOOLEAN;
  T : constant BOOLEAN := True;     F : constant BOOLEAN := False;
  A : BOOLEAN_VECTOR_t (1 .. 4) := (T, F, T, F);
  B : BOOLEAN_VECTOR_t (1 .. 4) := (T, F, F, T);
  C : BOOLEAN_VECTOR_t (1 .. 4);
begin
  C := not A;     -- yields (F, T, F, T);
  C := A and B;   -- yields (T, F, F, F);
  C := A or B;    -- yields (T, F, T, T);
  C := A xor B;   -- yields (F, F, T, T);
end;
```

Problem 2.5: Using type BOOLEAN_VECTOR, write an Ada program which generates truth tables for and, or, and xor.

2.6 ARRAY ATTRIBUTES AND OPERATIONS

Ada allows attributes associated with various kinds of objects to be accessed in a general-purpose fashion by using predefined attribute names attached to object names with a tick (single quote). An array composite data type has four attributes of particular interest:

1. FIRST - the value of the first index value
2. LAST - the value of the last index value
3. RANGE - array'FIRST .. array'LAST
4. LENGTH - the number of elements in an array

For example:

```
type FA_t is array (INTEGER range <>) of FLOAT;
X : FA_t (2 .. 5);
I : INTEGER;

I := X'FIRST;    -- I = 2
I := X'LAST;     -- I = 5
for J in X'RANGE loop  -- J goes from 2 to 5
  null;
end loop;
I := X'LENGTH;  -- I = 4
```

These attributes apply to array objects (which are, of course, constrained) and constrained array types. Operations on arrays are:

Figure 14 Array operations

Operation	Restriction
Attributes (FIRST, etc.)	None
Logical (not, and, or, xor)	Arrays must be BOOLEAN vectors of the same length and type
Concatenation (&)	The arrays must be vectors (one-dimensional)
Assignment (:=)	The arrays must be of the same size and type
Type Conversions	The arrays must be of the same size and component and index types
Relational (<, >, <=, >=)	The arrays must be discrete vectors of the same type
Equality (=, /=)	The arrays must be of the same type

Problem 2.6: Write an Ada program which contains a procedure that displays the values of an array of integer numbers passed to it. The procedure should use the array attributes to determine the index range of the array. Pass at least two arrays of different index ranges to this procedure.

2.7 RECORDS

A record is a composite data type in which the components of the record are not necessarily homogenous (i.e., of the same type, as in an array). A record type may be composed of a float, a character string, and an integer, for example, where an array type may be composed of just integers or just floats. Records are often the basis for abstract data types in Ada, and abstract data types are the basis for the definition of object member data in an object-based design written in Ada.

2.7.1 Records without Discriminants

The most basic kind of record is that declared without discriminants (parts of a record definition that allow one instance of a record to vary from another, thereby supporting variant records). The general syntax of a record type declaration without discriminants is:

```
type RECORD_TYPE_NAME is
  record
    RECORD_COMPONENTS;
  end record;
```

where RECORD_COMPONENTS is of the form

```
FIELD_NAME : TYPE_NAME [:= INITIAL_VALUE];
```

When an object of a record type is created, the fields of the object are accessed by using a dot notation of the form:

```
OBJECT_NAME.FIELD_NAME
```

If a field is a record itself, the dot notation may be continued. Examples of a record type and object definition and assignments are:

```
type MY_RECORD_t is
  record
    I : INTEGER;
    F : FLOAT := 5.0;
  end record;
```

```
X : MY_RECORD_t;  -- X.I is undefined, and X.F is initialized to 5.0
X : MY_RECORD_t := (I => 2);  -- X.I is initialized to 2, X.F is
                              -- initialized to 5.0
X : MY_RECORD_t := (I => 2, F => 4.0);  -- X.I = 2, X.F = 4.0
X : MY_RECORD_t := (2, 4.0);  -- same as above
```

2.7.2 Records with Discriminants

Record types with discriminants may be used to define records to be of the same type even though the kind, number, and size of the components differ between individual instances. Variant records are those that differ from one another in the kind and number of components. The variable component, which is just another component in the record, is specified in the type declaration like a parameter. An example of a variant record in Ada:

```
type RECORDING_MEDIUM_t is (PHONOGRAPH, CASSETTE, CD);
type MUSIC_TYPE_t is (CLASSICAL, JAZZ, NEW_AGE, FOLK, POP);
type RPM_t is (RPM_33, RPM_45);

type RECORDING_t (Device : RECORDING_MEDIUM_t := CD) is
  record
    Music : MUSIC_TYPE_t;
    case Device is
      when PHONOGRAPH => Speed  : RPM_t;
      when CASSETTE   => Length : NATURAL;
      when CD         => null;
    end case;
  end record;

My_CD       : RECORDING_t := (Music => POP);
His_Record  : RECORDING_t(PHONOGRAPH) := (Music => JAZZ, Speed => RPM_33);
My_Tape     : RECORDING_t(CASSETTE) := (Music => CLASSICAL, Length => 60);
```

2.7.3 Problem

> Problem 2.7: Write an Ada program which manipulates an object of type BOOK, where the BOOK class of objects consists of a character string for the title of the book, a character string for the author, a floating point number for the price, and an integer for the publication year. Create some objects of type BOOK, initialize them, and display their values.

2.8 ACCESS TYPES

Access types are used to declare variables (pointers) that access dynamically allocated variables. A dynamically allocated variable is brought into existence by an allocator (the keyword new). Dynamically allocated variables are referenced by an access variable, where the access variable "points" to the variable desired. The general form of an access type declaration is:

```
type ACCESS_TYPE_NAME is ACCESS TYPE_NAME;
```

A variable of an access type is the only kind of variable in Ada which is initialized without the programmer having to do anything. All variables of an access type are initialized to null (another keyword) when they are created.

With variables of an access type, the special suffix .all refers to the value of the variable being accessed. If the variable accessed is a record, the record fields may be addressed by using the same dot notation as used for a conventional record variable.

Examples:

```
type INTEGER_ACCESS_t is access INTEGER;
type RECORD_t is
  record
    I : INTEGER;
    F : FLOAT;
  end record;
type RECORD_ACCESS_T is access RECORD_t;

P1, P2 : INTEGER_ACCESS_t;
R1 : RECORD_ACCESS_t;
P1 := new INTEGER;  P1.all := 5;
P2 := new INTEGER'(12);  -- allocate and initialize
R1 := new RECORD_t;
R1.I := 12; R1.F := 2.2;
R1 := new RECORD_t'(I => 10, F => 10.0);
```

Problem 2.8: Write an Ada program which manipulates objects of a record type that contains a NATURAL indicating the number of valid elements in an array of integer numbers and an array of up to 100 integer numbers. This program should contain a procedure which receives an access type that addresses one of these objects and prints out the valid numbers. Have the mainline call this procedure two or more times with different arguments.

2.9 OBJECT REPRESENTATION

There are several attributes associated with type declarations and object definitions in Ada which can be used to direct affect the internal implementation details of these types and objects. The following are attributes which may be applied to various entities in order to determine some of their specific properties:

Figure 15. Object Attributes

Attribute	Description
ADDRESS	reports the memory location of an object, program unit, label, or task entry point
SIZE	reports the size, in bits, of an object, type, or subtype
STORAGE_SIZE	reports the amount of available storage for access types and tasks; if P is an access type, P'STORAGE_SIZE gives the amount of space required for an object accessed by P; if P is a task, P'STORAGE_SIZE give the number of storage units reserved for task activation
POSITION (records only)	reports the offset, in storage units, of a record component from the beginning of a record
FIRST_BIT (records only)	reports the number of bits that the first bit of a record component is offset from the beginning of the storage unit in which it is contained
LAST_BIT (records only)	reports the number of bits that the last bit of a record component is offset from the beginning of the storage unit that contains the first bit of the record component

These attributes only return information about the associated entities. In addition to these attributes, the following representation clauses (also known as representation specifications or rep specs) are available in Ada to set certain attributes:

- Length clauses -- establish the amount of storage used for objects

```
type DIRECTION_t is (UP, DOWN, RIGHT, LEFT);
for DIRECTION_t'SIZE use 2;  -- 2 bits
```

- Enumeration clauses -- specify the internal representation of enumeration literals

```
type BIT_t is (OFF, ON);
for BIT_t'SIZE use 1;   -- 1 bit
for BIT_t use (OFF => 0, ON => 1);
```

- Record Representation clauses -- associated record components with specific locations in bit fields

```
type STATUS_WORD_t is
  record
    Carry_Bit     : BIT_t;
    Overflow_Bit  : BIT_t;
    Fill_1        : BIT_t;
    Fill_2        : BIT_t;
    Fill_3        : BIT_t;
    Fill_4        : BIT_t;
    Fill_5        : BIT_t;
    Zero_Bit      : BIT_t;
  end record;
for STATUS_WORD_t'SIZE use 8;   -- 8 bits
for STATUS_WORD_t use
   record
    Carry_Bit     at 0 range 0..0;   -- bit 0
    Overflow_Bit  at 0 range 1..1;   -- bit 1
    Fill_1        at 0 range 2..2;   -- bit 2
    Fill_2        at 0 range 3..3;   -- bit 3
    Fill_3        at 0 range 4..4;   -- bit 4
    Fill_4        at 0 range 5..5;   -- bit 5
    Fill_5        at 0 range 6..6;   -- bit 6
    Zero_Bit      at 0 range 7..7;   -- bit 7
   end record;
```

- Address clauses -- specify the addresses of objects

```
  CPU_STATUS : STATUS_WORD;         -- define object
  for CPU_STATUS use at 16#100#;    -- define address of object
```

Note that Address clauses may also be apply to task entry points to establish interrupt mappings:

```
  task RUNNING_SCORE is
    entry HIT;  for HIT use at 16#020#;
    entry MISS; for MISS use at 16#040#;
  end RUNNING_SCORE;
```

Problem 2.9: Write an Ada program which extends the definition of STATUS_WORD given above, including a field which is more than one BIT long, and sets the values of several STATUS_WORD objects. Display the values of these objects.

2.10 OPERATORS

The following operators are supported in the Ada language:

Figure 16. Operators

Notes	Operators						Precedence
Exponentation, not and abs	**	not	abs				Highest
Multiply Operators	*	/	mod	rem			
Unary Operators	+	-					
Binary Operators	+	-	&				
Relational Operators	=	/=	<	<=	>	>=	
Membership Operators	in	not in					
Logical Operators	and	or	xor				
Short-Circuit Operators	and then		or else				Lowest

Figure 17. Sample expression using these operators

```
Expression                  Notes
--------------------        ----------------------------
PI                          a simple expression

(B**2) - (4.0 * A * C)

B**2 - 4.0 * A * C          same meaning as the above

CH in 'a' .. 'z'            a boolean expression

24.2**3                     a static expression

(not SUNNY) or WARM         a boolean expression

not SUNNY or WARM           same meaning as the above

not (SUNNY or WARM)         different from above

"Hello" & "  " & "Joe"      string concatenation

B > 0 and then A/B < 5      short circuit boolean expression
```

Problem 2.10: Write an Ada program which inputs two integer numbers and calculates their quotients, using a NATURAL for the quotient. Use short circuit operators to check for potential errors before attempting the divides.

3 ADA STATEMENTS

A statement in Ada is a sequence of characters terminated by a semicolon (;). For example,

```
Value := Value + 1;  -- an assignment statement

Value :=
2
;  -- another assignment statement

Value := 2;  -- same as the last assignment statement

if A < 4 then
  A := 4;
end if;  -- an if statement which contains one assignment statement
```

There are four broad categories of statements in Ada:

1. Sequential Control Statements
   - Assignment
   - Block
   - Null
   - Return
   - Procedure Call

2. Conditional Control Statements
   - Case
   - If

3. Iterative Control Statements
   - Exit
   - Loop (including For Loop and While Loop)

4. Other Statements
   - Abort
   - Accept
   - Code
   - Delay
   - Entry Call
   - Goto
   - Raise
   - Select

3.1 SEQUENTIAL CONTROL STATEMENTS

An assignment statement changes the value of a variable. The colon-equals operator (:=) is used to perform the assignment, as opposed to the equals operator (which is a test for equality, resulting in a boolean value of True or False). Examples of assignment statements:

```
Value := 1;
Value := SQRT(B**2 + A**2);
```

A block statement declares a logically-cohesive body of code with, optionally, its own local variables. In essence, a block is like an inline procedure with no parameters. The variables available within a block are all the variables in the subprogram which contains the block plus its own local variables. The general syntax of a block statement is:

```
LABEL:  -- optional block label
  declare  -- optional local variable declaration region
   -- local variables are defined here
  begin -- marks the beginning of the code in the block
```

```
    null;  -- local statements
  end LABEL;  -- end of block
```

An example of a block is:

```
declare
  Local_1 : INTEGER;
begin
  Local_1 := 2;
  Value := Value / Local_1; -- the variable VALUE is global
                            -- to the block
end;
```

The null statement is simply the statement

null;

and it is usually employed as a placeholder during design to mark where code will go in the future while still allowing the design to be compiled.

The return statement is used to return from a subprogram. If the subprogram is a procedure, the form of the return statement is:

return;

and if the subprogram is a function, the form of the return statement is: **return value**; for example:

```
procedure X is
begin -- X
  return;
end X;

function Y return FLOAT is
begin -- Y
  return 1.0;
end Y;
```

A procedure call statement is simply a call to a procedure. The procedure name may be prefixed with the name of a containing program unit, such as a package, if necessary. The general form of a procedure call:

```
PROCEDURE_NAME;  -- no arguments

PROCEDURE_NAME(ARG, ARG, ...);  -- arguments

PACKAGE_NAME.PROCEDURE_NAME (ARG, ARG, ...); -- procedure in a package
```

Examples:

```
Text_IO.Put_Line("Hello, world");
Put("Enter Text: ");
Stacks.Push(100.0, My_Stack);
```

Problem 3.1: Write an Ada program which contains a procedure and a block with local variables. Place the procedure inside the block and call it several times.

3.2 CONDITIONAL CONTROL STATEMENTS

There are two basic kinds of conditional control statements in Ada: the if statement and the case statement.

The if statement is of the following general form:

```
if BOOLEAN_EXPRESSION then
  -- statements to execute if true
elsif ANOTHER_BOOLEAN_EXPRESSION then -- alternate if
  -- statements to execute if true
elsif YET_ANOTHER_BOOLEAN_EXPRESSION then -- as many as desired
  -- statements to execute if true
else -- what to do if all else fails
  -- statements to execute if all are false
end if;  -- end of if statement
```

Example of the if statement:

```
type COLOR_t is (RED, GREEN, YELLOW);
Stop_Light : COLOR_t := RED;
if Stop_Light = RED then
  Stop;  -- procedure call
elsif Stop_Light = YELLOW then
  Close_Eyes; Go_Fast;  -- two procedure calls
else  -- must be GREEN
  Stop;
  Look_Both_Ways;
  Go;  -- three procedure calls
end if;

if Value > 10 then
  Value := Value - 10;
end if;
```

The case statement is a multi-way selection. The general form of the case statement is:

```
case VARIABLE_NAME is
  when VARIABLE_VALUE =>
    -- statements to perform when variable_name = variable_value
  when VARIABLE_VALUE2 =>
    -- and so on
  when others =>
    -- statements to perform when variable_name
    -- takes on any other value
end case;
```

Every possible value of the case variable name must be covered in one and only one `when` clause. The `when others` clause is used to cover all instances not explicitly covered before and always appears as the last `when` clause. Choices in a when clause must be static (able to be resolved at compile time). Case variable names must be discrete.

Examples of case statements:

```
case Value is
  when 1 | 3 | 5 | 7 | 9 => Kind := ODD;
  when others => Kind := EVEN;
end case;

case Value is
  when 0 .. 9 => Kind := LESS_THAN_10;
  when others => Kind := TEN_OR_MORE;
end case;
case Stop_Light is
  when RED =>
    Stop;
  when GREEN =>
    Look_Both_Ways;
    Go;
  when YELLOW =>
    Close_Eyes;
    Go_Fast;
  when others =>  -- off?  but we have already covered all
                  -- the enumeration values
    Stop;
    Look_Both_Ways;
    Go;
  end case;
```

Problem 3.2: Write a function which acts on a CHARACTER and returns an enumeration value of the set (LOWER_CASE, UPPER_CASE, SHARP, COMMA, SEMICOLON, DIGIT, SOME_OTHER). Write an Ada mainline which calls this function with a number of different characters and displays the character and the result of the function call.

3.3 ITERATIVE CONTROL STATEMENTS

There are two kinds of exit statements and three kinds of loop statements which make up Ada's iterative control statements.

The exit statements cause a premature exit from a loop or a block. The two kinds of exit statements are:

```
exit;  -- unconditional
exit when boolean_expression;  -- conditional
```

Examples:

```
for I in 1 .. 10 loop
  if I = 5 then
    exit;
  end if;
end loop;

for I in 1 .. 10 loop
  exit when I=5;
end loop;
```

The loop statements cause iterations to occur. The three kinds of loop statements are:

1. Simple loop:

```
OPTIONAL_LABEL:
loop
  statements;
end loop OPTIONAL_LABEL;
```

2. While loop:

```
OPTIONAL_LABEL:
while BOOLEAN_EXPRESSION loop
  statements;
end loop OPTIONAL_LABEL;
```

3. For loop:

```
OPTIONAL_LABEL:
for LOOP_VARIABLE in LOOP_VALUE_RANGE loop
  statements;
end loop OPTIONAL_LABEL;
```

Examples of loop statements:

```
loop -- simple loop
  Bit := Status_Bit;
  exit when Bit = ON;
end loop;

while Status_Bit = OFF loop
  null;  -- nothing for now
```

```
end loop;

I := 42;
ADD_UP:
for I in 1..20 loop  -- for loop, creates local I variable and
                     -- hides outer I variable
  Sum := Sum + I;
end loop ADD_UP;
Sum := Sum + I;      -- add in outer I variable
```

Problem 3.3: Write an Ada program which loops up to 10 times, accepting a string input from the user and exiting if the string contains the word "EXIT" in the first four characters.

4 ADA PROGRAM UNITS

Program units in Ada are the basic components of Ada libraries. When an Ada compiler compiles an Ada source file, it places the programs units in this source file which it compiles successfully into a "current" program unit library. When an Ada system is completely compiled, an executable is built by a binder starting with a procedure program unit that has no parameters and including all the program units it requires, all the program units they require, and so on until no more program units are needed. An Ada runtime system or an interface to an Ada runtime system is added to this collection of program units, and the binder places all of this into a single executable file.

There are four kinds of program units in Ada:

1. subprograms (procedures and functions)
2. packages
3. generic units
4. tasks

4.1 SUBPROGRAMS (AND BLOCKS)

A subprogram in Ada is a unit of sequential action, usually used to encapsulate the performance of a single logical "operation" with an optional set of parameters. A block in Ada is a subunit of sequential action found within a program unit like a subprogram. There are two kinds of subprograms: procedures (which return zero or more values) and functions (which return only one value). Blocks, procedures, and functions contain three parts:

1. an optional declarative part, in which local variables are defined

2. an executable statement part, in which code resides

3. an optional exception handler, in which code to handle exceptional conditions (such as divide by zero) resides

The declarative part contains declarations of types and subtypes, variables and constants, and encapsulated program units (subprograms, packages, generic units, and tasks). The entities brought into existence in the declarative part only exist as long as the block, procedure, or function in which they reside is active.

The executable statement part contains executable statements, such as assignment and control statements.

The exception handler traps error conditions, or exceptions, and processes them.

Subprograms are the basic program units employed to perform sequential action in an Ada system. There are two classes of subprograms:

1. `procedures` - accept and return values in parameter lists

2. `functions` - accept values in parameter lists and only return one value

The optional parameter lists contain three classes of formal parameters:

`in` - parameter values are passed into subprograms (internally, these parameters may only be used on the right-hand side of an assignment statement)

`out` - parameter values are passed out of subprograms (internally, these parameters may only be used on the left-hand side of an assignment statement); `out` parameters may be used in procedures only

`inout` - parameter values are passed both ways; `inout` parameters may be used in procedures only

Overloading Subprograms. Subprogram names may be overloaded (i.e., two or more subprograms may have the same name but different types or numbers of parameters), and Ada can resolve these from context.

Recursion in Subprograms. A subprogram may call itself, or recurse.

The Basic Differences Between Subprograms and Blocks. Subprograms can be compiled separately, while blocks are embedded in some larger unit which must be compiled as a whole. Embedded subprograms can only be placed in the declarative part of a program unit, while blocks can only be placed in the executable statement part. Subprograms can be invoked by a call, while blocks are invoked as part of the flow of execution only.

4.1.1 Blocks

The general form of a block is:

```
OPTIONAL_LABEL:
declare -- optional
  -- variable definitions
begin
  null;  -- statements
exception -- optional
  when others => null;  -- exception handler
end OPTIONAL_LABEL;
```

Example:

```
MY_BLOCK:
declare
  I : INTEGER := 5;
begin
  Value := Value / I;  -- Value is an INTEGER external to the block
end MY_BLOCK;
```

4.1.2 Function Subprograms

The general syntax of a function is:

```
function FUNCTION_NAME ( PARAMETERS ) return TYPE; -- specification
function FUNCTION_NAME ( PARAMETERS ) return TYPE is -- body
  -- optional variable definitions
begin -- FUNCTION_NAME
  -- statements (including at least one return statement)
exception -- optional
  when others => null;  -- exception handlers
end FUNCTION_NAME;
```

Examples of function specifications and bodies:

```
function Sin(Angle : in FLOAT) return FLOAT:  -- spec

function Sin(Angle : in FLOAT) return FLOAT is -- body
begin -- Sin
  null;  -- detail omitted
  return 1.0;  -- dummy return value
end Sin;

function Cos(Angle : FLOAT) return FLOAT;  -- mode is 'in' by default
function "*" (Left, Right : in COMPLEX_NUMBER) return COMPLEX_NUMBER;
  -- operator overloading
```

Examples of function calls:

```
X  := Sin(2.2);
X  := Cos(Angle => 45.2);
Y  := Trig_Lib.Cos(X);
C3 := Complex."*"(C1, C2);  -- prefix call
C3 := C1 * C2;  -- infix call, same as above prefix call
```

4.1.3 Procedure Subprograms

The general syntax of a procedure is:

```
procedure PROCEDURE_NAME ( PARAMETERS );  -- specification
procedure PROCEDURE_NAME ( PARAMETERS ) is  -- body
  -- optional local variables
begin -- PROCEDURE_NAME
  -- statements (a return statement is not required)
exception -- optional
  when others => null;  -- exception handler
end PROCEDURE_NAME;
```

Examples of procedure specifications and bodies:

```
procedure Do_It;  -- specification

procedure Get_Status(Result : out STATUS);    -- spec
procedure Get_Status(Result : out STATUS) is -- body
begin -- Get_Status
  Result := OK;
end Get_Status;

procedure Create(File : in out FILE_TYPE;
                 Name : in STRING := "DUMMY.TXT";
                 Mode : in FILE_MODE := IN_FILE);
-- spec with default values for parameters
```

Examples of calls to the above procedures:

```
Do_It;
Get_Status(Value);
Get_Status(Result => Value);

Create (FD);  -- file name is "DUMMY.TXT" and mode is IN_FILE
Create (FD, Mode => OUT_FILE);  -- file name is still "DUMMY.TXT"
Create (File => FD,
        Name => "Myfile.txt",
        Mode => IN_FILE);
```

4.1.4 Problem

> Problem 4.1: Write an Ada program which defines the type UNSIGNED_BYTE and functions "+", "-", "*", and PRINT which operate on objects of type UNSIGNED_BYTE. Illustrate the use of these functions as infix operators when possible.

4.2 PACKAGES

A package is an encapsulation mechanism in Ada, allowing the programmer to collect groups of entities together. As a rule, these entities should be logically related. A package usually consists of two parts, like the other program units: a specification and a body.

Packages directly support object-oriented programming, providing a means to describe a class or object (an abstract data type). A package may implement either an object as an abstract state machine (wherein the state information is stored as data hidden in the body of the package) or a class (an abstract data type) using private data types (wherein the state information is stored as private data associated with each object).

Packages are used for four main purposes:

1. as collections of constants and type declarations
2. as collections of related functions
3. as abstract state machines
4. as abstract data types

Package bodies may contain an optional initialization part. If this is present, the code of the initialization part of a package is executed before the first line of code in the mainline procedure of an Ada system.

Packages may be embedded in blocks, subprograms, and any program unit in general.

A private type is a type declaration which is visible in the specification of a package, but its underlying implementation is hidden from the code withing the package and is of no concern to the outside world. Private types are the means of implementing abstract data types in Ada. In a package containing a private type, the only operations which may be performed on objects of that type are assignment, tests for equality and inequality, and the procedures and functions explicitly exported by the package. In a package containing a limited private type, the only operations which may be performed on objects of that type are the procedures and functions explicitly exported by the package.

The general form of a package specification is:

```
package PACKAGE_NAME is
  type TYPE_NAME is private;
    -- optional reference to a private type
    -- visible declarations
private
  -- private type declarations
end PACKAGE_NAME;
```

The general form of a package body is:

```
package body PACKAGE_NAME is
  -- implementations of code and hidden data
begin -- optional
  -- initialization statements
end PACKAGE_NAME;
```

Examples of package specifications:

```
package Console is -- abstract state machine

  type LOCATION_t is
    record
      Row : INTEGER;
      Col : INTEGER;
    end record;

  procedure Clear_Screen;
  procedure Position_Cursor(Where : in LOCATION);
  procedure Write(Item : in CHARACTER);
  procedure Write(Item : in STRING);
```

```
  end Console;

  package Complex is -- abstract data type

    type OBJECT_t is private;

    function Set(Real : in FLOAT; Imag : in FLOAT) return OBJECT_t;
    function Real_Part(Item : in OBJECT_t);
    function Imag_Part(Item : in OBJECT_t);
    function "+"(Left, Right : in OBJECT_t) return OBJECT_t;
    function "-"(Left, Right : in OBJECT_t) return OBJECT_t;

private

  type OBJECT_t is
    record
      RP : FLOAT;
      IP : FLOAT;
    end record;

  end Complex;
```

Problem 4.2: Write an Ada system containing a package UNSBYTE which includes the operations in problem 4.1. Set up package UNSBYTE as an abstract data type.

4.3 GENERIC UNITS

Generic subprograms and packages, which are templates describing general-purpose algorithms that apply to a variety of types of data, may be created in Ada systems. The bodies of generic functions and procedures resemble normal subprograms except that the general types used in the specifications are employed rather than conventional types. These generic units describe the algorithms in general terms, specifying what kinds of restrictions must be placed on the data to which these templates apply. For example, certain algorithms require that the elements must have an associated test for equality and must be able to be assigned. Other algorithms require that the elements resemble integers or floating point numbers. The generic formal parameters presented in a specification of a generic unit explicitly identify the generic parameters and the restrictions placed upon them.

Generic functions look like this:

```
generic
  -- generic formal parameters
function FUNCTION_NAME ( PARAMETERS ) return TYPE;  -- spec
```

Generic procedure look like this:

```
generic
  -- generic formal parameters
procedure PROCEDURE_NAME ( PARAMETERS );  --spec
```

Generic packages look like this:

```
generic
  -- generic formal parameters
package PACKAGE_NAME is  -- spec
  -- normal package stuff
end PACKAGE_NAME;
```

An example of a full generic specification and body:

```
generic
  type ELEMENT_t is private;  -- anything that can be assigned

procedure Exchange(Item1, Item2 : in out ELEMENT_t);  -- spec

procedure Exchange(Item1, Item2 : in out ELEMENT_t) is  -- body
  Temp : ELEMENT_t;   -- temporary is the same type as the parameters
begin
  Temp  := Item1;
  Item1 := Item2;
  Item2 := Temp;
end Exchange;
```

The generic formal parameter(s) must be specified during the instantiation, specifying what external, existing entities are to corresponding to those generic parameters. An example:

```
type X_t is
  record
    I : INTEGER;
    F : FLOAT;
  end record;

procedure Int_Exchange is new Exchange (ELEMENT => INTEGER);
procedure Float_Exchange is new Exchange (FLOAT);
procedure X_Exchange is new Exchange (X_t);

I1, I2 : INTEGER;
F1, F2 : FLOAT;
X1, X2 : X_t;
```

```
Int_Exchange(I1, I2);  -- actually use the new procedures
Float_Exchange(F1, F2);
X_Exchange(X1, X2);
```

Hence, the generic formal parameters are the key to understanding, writing, and using generic units. There are three kinds of generic formal parameters: types, objects, and subprograms.

1. Types as generic formal parameters:

Figure 18. Types as generic formal parameters:

Type Parameter	Operations Allowed	Applicable Data Types
type T_t is private;	= /= :=	All assignable
type T_t is limited private;	- none -	All
type D_t is (<>);	= /= := > >= < <= PRED SUCC FIRST LAST	Discrete
type I_t is range <>;	Integer operations	Integer
type F_t is digits <>;	Real operations	Float
type FIXED_t is delta <>;	Fixed point operations	Fixed

2. Object declarations may appear as formal parameters.

3. Subprograms may appear as formal parameters.

Problem 4.3: Write an Ada program which includes a generic procedure which finds the index of the lowest item in a list of Items. Instantiate this package to work with arrays of integers, floats, and records. Display the arrays and the item with the lowest (smallest) value.

4.4 TASKS

In Ada, one can write programs that perform more than one activity concurrently. This concurrent processing is called tasking, and the units of code that run concurrently are called tasks.

A simple format for task specifications and bodies:

```
task TASK_NAME;  -- specification
task body TASK_NAME is -- body
  -- local variable declarations
begin
  -- code
end TASK_NAME;
```

A more complex format:

```
task TASK_NAME is -- specification
  entry entry_name ( parameters );
end TASK_NAME;

task body TASK_NAME is  -- body
  -- local variables
begin
  accept ENTRY_NAME ( PARAMETERS ) do -- code follows
    -- code at entry point
  end ENTRY_NAME;
end TASK_NAME;
```

The entry statement in the task specification identifies the entry points to the task. The accept statement in the task body identifies the code to be executed at that entry point. The entry points to a task are called like subprogram calls from other program units.

A task type may be created as well:

```
task type TASK_NAME1_t;  -- specification
task body TASK_NAME1_t is -- body
  -- local variable declarations
begin
  -- code
end TASK_NAME1_t;

task type TASK_NAME2_t is -- specification
  entry ENTRY_NAME ( PARAMETERS );
end TASK_NAME2_t;

task body TASK_NAME2_t is  -- body
  -- local variables
begin
  accept ENTRY_NAME ( PARAMETERS ) do -- code follows
    -- code at entry point
  end ENTRY_NAME;
end TASK_NAME2_t;
```

When we have task types, we have the ability to statically or dynamically create task objects:

```
My_task : TASK_NAME1_t;  -- task is running right after definition
Task_vector : array (1 .. 20) of Task_name2_t;  -- array of tasks
```

The interfacing of two tasks in order to pass data is called a rendezvous in Ada. Entry points are called like subprograms to effect a rendezvous.

> Problem 4.4: Write an Ada system that consists of a task type that accepts a name string and an amount for the task to delay, delays the indicated amount, and displays a message when done. In your mainline, create and start an array of 5 tasks of this type.

4.5 EXCEPTIONS

Two kinds of errors are commonly encountered in programming: compilation errors and runtime errors. In Ada, runtime errors are called exceptions. Exceptions may be predefined or user-defined. To define an exception:

```
Exception_name : exception;
```

To raise an exception to be handled by an exception handler somewhere in the calling chain:

```
raise Exception_name;
```

Exception handlers are Ada constructs that handle exceptions. An exception handler is placed at the end of a block, subprogram, package, or task, and is denoted by the keyword exception followed by the text of the handler code. Example (for a block):

```
begin -- note that I is defined external to this block

  I := I / 0;  -- division by zero
exception
  when NUMERIC_ERROR =>
    I := 10;
end;
```

Predefined Exceptions. There are two main sources for predefined exceptions in Ada: package STANDARD and package TEXT_IO:

1. Package `STANDARD`

   - `CONSTRAINT_ERROR` - raised whenever a value goes out of bounds, such as assigning a value of 11 to a variable whose type is in the range from 0 to 10

   - `NUMERIC_ERROR` - raised when illegal or unmanageable mathematical operations are performed, such as dividing by zero

   - `STORAGE_ERROR` - raised when the computer runs out of available memory

   - `TASKING_ERROR` - raised when there is a problem with the multitasking environment, such as calling a task which is no longer active

   - `PROGRAM_ERROR` - all other exceptions not covered by the above or other exceptions defined by the user or some other package, such as reaching the end of a function without hitting a return statement

2. Package `TEXT_IO`

   - `DATA_ERROR` - encountered an input that is not of the required type

   - `DEVICE_ERROR` - malfunction in the underlying system, such as disk space being full

   - `END_ERROR` - an attempt was made to read past the end of the file

   - `LAYOUT_ERROR` - raised by COL, LINE, or PAGE if the value returned exceeds `COUNT'LAST`

   - `MODE_ERROR` - an attempt was made to read from or test the end of a file whose current mode is `OUT_FILE`, or an attempt was made to write to a file whose current mode is `IN_FILE`

   - `NAME_ERROR` - the string given as a file name to `CREATE` or `OPEN` does not allow the identification of an external file

   - `STATUS_ERROR` - an attempt was made to operate on a file that is not open or open a file that is already open

   - `USE_ERROR` - an operation is attempted that is not possible for reasons that depend on the characteristics of the external file

Exception Propagation. If the program unit that raises an exception does not contain an exception handler that handles the exception, the exception is propagated to the next level beyond the unit. This level varies, depending on the unit raising the exception:

- If the unit is a mainline procedure, the Ada runtime environment handles the exception by aborting the program.

- If the unit is a block, the exception is passed to the program unit (or block) containing the block that raised the exception.

- If the unit is a subprogram, the exception is passed to the program unit or block that called the subprogram.

The propagation path of an exception is determined at runtime.

To reraise the current exception in an exception handler, the statement raise; may be used.

Suppressing Exceptions. Ada performs many checks at runtime to ensure that array indices are not exceeded, variables stay within range, etc. If these checks fail, exceptions are raised. This results in larger code and slower execution speed in general.

In certain real-time applications, where space and time constraints are critical, runtime error checking may be too expensive. A solution is to use exception suppression.

Exception suppression turns off runtime error checking. It is implemented by a pragma (a compiler directive) called SUPPRESS:

```
pragma SUPPRESS (RANGE_CHECK);
  -- turns off range checking on array indices and variable values
pragma SUPPRESS (RANGE_CHECK, INTEGER);
  -- turns off range checking on integers only
pragma SUPPRESS (RANGE_CHECK, X);
  -- turns off range checking for a particular object
```

Problem 4.5: Write an Ada system that contains its own exception, DIVIDE_BY_ZERO. Write a function "/" that divides two integers, checking first to see if the second integer is zero (in which case it raises DIVIDE_BY_ZERO). Your mainline enter a loop, dividing several integer pairs and printing the result. If the exception DIVIDE_BY_ZERO is raised, the mainline should print out that this happened, show the attempted division, and continue with the rest of the integer pairs.